Introduction à VHDL

Daniel Etiemble
de@lri.fr

Introduction à VHDL

- Exemple de conception
  - Code VHDL pour réaliser un comparateur 4 bits




```
-- eqcomp4 est un comparateur 4 bits
entity eqcomp4 is
  port (a, b: in bit_vector(3 downto 0);
        equals : out bit);
end eqcomp4;

architecture dataflow of eqcomp4 is
begin
   equals <= '1' when (a = b) else '0';
end dataflow;
```

NOTE:
1. -- indique un commentaire
2. Deux parties dans le code

Format VHDL

- Le code VHDL décrit une entité en deux parties
  - Déclaration d'entité
  - Corps de l'architecture
- Il y a aussi d'autres aspects
  - package
  - configuration



Déclaration d'entité

- Description des entrées sorties
  - Comme le schéma d'un composant
  - Description des "ports" d'un composant (les "pattes" d'E/S du schéma)
  - **entity** <name> **is** . . . **end** <name>;
  - EXEMPLE : un additionneur 4 bits



Schéma

```
entity add4 is port(
   a, b:  in std_logic_vector(3 downto 0);
   ci:    in std_logic;
   sum: out std_logic_vector(3 downto 0);
   co:    out std_logic);
end add4;
```

PORTS

- Chaque signal d'E/S d'une entité est appelé port
  - Un port est un objet de type données qui peut recevoir des valeurs et être utilisé dans des expressions
  - Chaque port a un nom, une direction (mode) et un type de données
- Nom de port
  - Majuscules/minuscules sont équivalents
  - Le premier caractère doit être une lettre
  - Le dernier caractère ne peut être un "underscore"
  - Deux "underscores" successifs sont interdits

Modes et types

- Un mode décrit la direction du transfert d'une donnée à travers un port
  - in
  - out
  - inout (signaux bidirectionnels)
  - buffer (un nœud interne utilisé pour feedback)
- Il y a plusieurs types de données disponibles
  - Boolean, bit, bit_vector, integer
  - La bibliothèque standard IEEE fournit également std_ulogic et std_logic et des tableaux de ces deux types
    - On doit signaler à VHDL d'utiliser cette bibliothèque

```
library ieee;
use iee.std_logic_1164.all;
```

Architecture

- Une architecture décrit le contenu d'une entité
- VHDL a trois styles d'architecture qui peuvent être combinée dans le corps d'une architecture
  - Comportemental
  - Flot de données
  - Structurel
- Le même circuit peut être décrit en utilisant n'importe lequel des trois styles

Description comportementale

```
library ieee;
use ieee.std_logic_1164.all;
entity eqcomp4 is port(
  a, b:     in std_logic_vector(3 downto 0)
  equals:   out std_logic);
end eqcomp4;

architecture behavioral of eqcomp4 is
begin
  comp:  process (a, b)
         begin
           if a = b then
             equals <= '1';
           else
             equals <= '0';
           end if;
         end process comp;
end behavioral;
```

- Une description comportementale fournit un algorithme qui modélise le fonctionnement du circuit
- Une déclaration de processus contient un algorithme
  - Elle commence par une étiquette (optionnelle), le mot clé "process" et une liste des signaux actifs (*sensivity list*)
  - La liste des signaux actifs indique quels signaux provoqueront l'exécution du processus

Description flot de données

- Une description flot de données spécifie comment la donnée est transférée de signal à signal sans utiliser d'affectations séquentielles (comme dans la description comportementale)
- Cette description ne nécessite pas de déclaration de processus
- Toutes les déclarations s'exécutent en même temps (*"concurrent signal assignment"*)

```
architecture dataflow of eqcomp4 is
begin
    equals <= '1' when (a = b) else '0';
end dataflow;
```


Descriptions structurelles

```
library ieee;
use ieee.std_logic_1164.all;
entity eqcomp4 is port(
  a, b:      in std_logic_vector(3 downto 0)
  equals:    out std_logic);
end eqcomp4;

use work.gatespkg.all;
architecture struct of eqcomp4 is
   signal x : std_logic_vector(0 to 3)
begin
   u0:  xnor2 port map (a(0),b(0),x(0));
   u1:  xnor2 port map (a(1),b(1),x(1));
   u2:  xnor2 port map (a(2),b(2),x(2));
   u3:  xnor2 port map (a(3),b(3),x(3));
   u4:  and4 port map (x(0),x(1),x(2),x(3),equals);
end struct;
```

- Les composants sont instanciés et connectés.
  - Ils doivent être définis dans un package et compilés dans une bibliothèque
  - Les bibliothèques sont attachées par une déclaration

xnor2 and and4 sont dans la bibliothèque work.gatespkg.

Modélisation structurelle en VHDL

- Concevoir une fonction majorité à trois entrées
  - La déclaration d'entité est équivalente à



  - La déclaration d'architecture est équivalente à



Les composants sont des portes ET et des portes OU

VHDL Code

- Aspect du code VHDL :
  - L'architecture a trois déclarations
    - Deux composants
    - Un signal
  - Les déclarations de composants correspondent à leur entité
  - Les déclarations de signaux créent un ou plusieurs signaux internes

```
entity MAJORITY is port(
   A_IN, B_IN, C_IN: in BIT;
   Z_OUT             : out BIT);
end MAJORITY;


architecture struct of MAJORITY is
--Declare logic operators
component AND2_OP
  port (A, B : in BIT; z : out BIT);
end component;
component OR3_OP
  port (A, B, C : in BIT; z : out BIT);
end component;
--Declare signals to interconnect logic operators
 signal INT1, INT2, INT3 : BIT;
begin
    A1:  AND2_OP port map (A_IN,B_IN,INT1);
    A2:  AND2_OP port map (A_IN,C_IN,INT2);
    A3:  AND2_OP port map (B_IN,C_IN,INT3);
    A4:  OR3_OP port map (INT1,INT2,INT3,Z_OUT);
end struct;
```

Déclaration de composants

- Les composants nécessaires sont aussi décrits avec un style VHDL

```
entity AND2_OP is
  port
    (A, B : in BIT;
     Z    : out BIT);
end AND2_OP;

architecture MODEL of AND2_OP is
begin
  Z <= A and B;
end MODEL;
```

```
entity OR3_OP is
  port
    (A, B, C : in BIT;
     Z       : out BIT);
end OR3_OP;

architecture MODEL of OR3_OP is
begin
  Z <= A or B or C;
end MODEL;
```

Ici modélisation "flot de données"

Instanciation des composants

- Le circuit Majorité contient quatre déclarations d'instanciation de composants
  - Elles commencent par une étiquette suivie de :
  - Puis un nom de composant
  - "port map" décrit comment chaque composant est connecté au reste du système.
- Ce sont des déclarations concurrentes : l'ordre n'a pas d'importance
  - Elles sont toutes exécutées en même temps

```
begin
  A1: AND2_OP port map (A_IN,B_IN,INT1);
  A2: AND2_OP port map (A_IN,C_IN,INT2);
  A3: AND2_OP port map (B_IN,C_IN,INT3);
  A4: OR3_OP port map (INT1,INT2,INT3,Z_OUT);
end struct;
```

Logique séquentielle

- Passage d'un état à l'état suivant sur une transition d'horloge

```
label: process (CLK)
        begin
                if(rising_edge(CLK)) then
                        D       <=      C;
                        B       <=      A;
                        C       <=      B;
                end if;--CLK
        end process;--main
```



Structures hiérarchiques

- Il n'y a pas de restrictions sur la hiérarchisation des modèles structurels
  - Les entités d'une réalisation peuvent utiliser des composants dont les entités à leur tour peuvent utiliser des composants

EXEMPLE : un circuit avec deux circuits majorité qui détecte quand deux groupes de trois entrées ont tous deux un signal de sortie haut.

```
entity MAJORITY_2x3 is port(
   A1,B1, C1    : in BIT;
   A2, B2, C2   : in Bit;
   Z_OUT        : out BIT);
end MAJORITY_2x3;

architecture struct of MAJORITY_2x3 is
component AND2_OP
  port (A, B : in BIT; z : out BIT);
end component;
component MAJORITY
  port (A_IN,B_IN,C_IN : in BIT
        Z_OUT          : out BIT);
end component;
 signal INT1, INT2: BIT;
begin
   M1:  MAJORITY port map (A1,B1,C1,INT1);
   M2:  MAJORITY port map (A2,B2,C2,INT2);
   A1:  AND2_OP port map (INT1,INT2,Z_OUT);
end struct;
```

PACKAGES

```
package LOGIC_OPS is
component AND2_OP
  port (A, B : in BIT; z : out BIT);
end component;
component OR32_OP
  port (A, B, C : in BIT; z : out BIT);
end component;
component NOT_OP
  port (A, : in BIT; A_BAR : out BIT);
end component;
end LOGIC_OPS;
```

- Un package sert à réutiliser un ensemble de déclarations de composants
  - Un package commence par le mot clé *package* et se termine par le mot clé *end*
  - Il contient les définitions de composants et de signaux (et plus…)
- Ce package pourrait être sauvegardé dans la bibliothèque WORK et appelé par
  - uses WORK.LOGIC_OPS.all
- Ou il pourrait être inclus dans le fichier source qui contient toutes les entités pour un problème

Bibliothèques

- Une bibliothèque contient du code compilé d'entités de conception
  - Il y a une bibliothèque standard VHDL
  - Lors de la compilation d'un projet, il est rangé par défaut dans la bibliothèque WORK
- L'utilisation d'une bibliothèque implique la déclaration

  library <nom_bibliothèque>;
  - Pour accéder à ses packages:

    use <nom_bibliothèque>.<nom_bibliothèque>.*all*;

Opérateurs pour instructions spécialisées NIOS

A spécifier à l'aide de VHDL (ou Verilog)



Les entités

UNAIRES

```
entity X is
       port
       (
       dataa : in    std_logic_vector(31 downto 0);
       result :out   std_logic_vector(31 downto 0)
       );
end X;
```

BINAIRES

```
entity Y is
       port
       (
       dataa : in    std_logic_vector(31 downto 0);
       datab : in    std_logic_vector(31 downto 0);
       result :out   std_logic_vector(31 downto 0)
       );
end Y;
```

Exemple d'architectures (1)

- Opérateurs MAX sur 32 bits

```
architecture comp of max32 is
begin
       result <= dataa when dataa > datab else datab ;
end comp; -- end of architecture
```

```
use IEEE.std_logic_unsigned.all;
--use IEEE.std_logic_signed.all;
```


Exemple d'architecture (2)

- MIN32 SIMD (4 x 8 bits)

```
architecture comp of min32simd is

begin
result (7 downto 0) <= dataa (7 downto 0)
         when (dataa (7 downto 0) < datab (7 downto 0))
         else datab (7 downto 0);
result (15 downto 8) <= dataa (15 downto 8)
         when (dataa (15 downto 8) < datab (15 downto 8))
         else datab (15 downto 8);
result (23 downto 16) <= dataa (23 downto 16)
         when (dataa (23 downto 16) < datab (23 downto 16))
         else datab (23 downto 16);
result (31 downto 24) <= dataa (31 downto 24)
         when (dataa (31 downto 24) < datab (31 downto 24))
         else datab (31 downto 24);
end comp; -- end of architecture
```